
1. Overview

2. Table of Contents

21. Overview

2. Table of Contents
2
3. Correction to PCell configuration in common CA functions
2
Change 1 – Correction to functions ‘f_EUTRA_508RRC_AddModRel_Scell_MIMO’, ‘f_EUTRA_ConfigPcell_AddScells’ and ‘f_EUTRA_508RRC_AddModRel_Scell_Common’
2

3. Correction to PCell configuration in common CA functions
Change 1 – Correction to functions ‘f_EUTRA_508RRC_AddModRel_Scell_MIMO’, ‘f_EUTRA_ConfigPcell_AddScells’ and ‘f_EUTRA_508RRC_AddModRel_Scell_Common’
	Function name
	f_EUTRA_508RRC_AddModRel_Scell_MIMO (),
f_EUTRA_ConfigPcell_AddScells (),
f_EUTRA_508RRC_AddModRel_Scell_Common ()

	Reason for change
	Under specific circumstances (like CIF or CQI configuration) the length of the DCI may change at CA configuration (… when the padding does not cover the increment of the DCI bits in certain Bandwidths). Since the PCell configuration is not timed in the common CA functions, it can happen that the DCI length changes prior to the RRC Reconfiguration and so the RRC message gets scheduled with a wrong DCI and the UE cannot decode the allocation.

	Summary of change
	It’s proposed to configure the PCell with an appropriate timing.

	TTCN module
	LTE_A_R10_R11\Common\EUTRA_CA\EUTRA_CommonFunctions_CA.ttcn

	MCC160 Comment
	Accepted as improvement


Before change

...
  function f_EUTRA_508RRC_AddModRel_Scell_MIMO(EUTRA_CellId_Type p_PCellId,
                                               EUTRA_CellId_Type p_SCellId,
                                               template (value) DL_DCCH_Message p_RRC_ConnRecongMsg,
                                               template (value) Scell_Capability_Type p_Scell_Capability := UL_DL,
                                               template (omit) MeasGapConfig p_MeasGapConfig := omit,
                                               template (omit)  DciUlInfo_Type p_DciUlInfo_SCell := cs_DciInfo_CcchDcchDtchUL_Def,
                                               template (value) UL_GrantConfig_Type  p_UL_GrantConfig_SCell := cs_UL_GrantConfig_Def)
    runs on EUTRA_PTC
  {
    var boolean v_MeasGapConfigIsPresent := isvalue(p_MeasGapConfig);
    var DrxCtrl_Type v_DrxCtrl;
    var RRCConnectionReconfiguration_r8_IEs v_RRCConnectionReconfigurationIEs := valueof(p_RRC_ConnRecongMsg.message_.c1.rrcConnectionReconfiguration.criticalExtensions.c1.rrcConnectionReconfiguration_r8);
    var SCellIndex_r10 v_SCellIndex := 1;
    var MAC_MainConfig_ScellDeactivationTimer_Type v_DeactTimer := rf32;
    var boolean v_NoUL_SCell ; //@sic R5s140083 sic@
    var template (omit) STAG_Id_r11 v_STAG_Id := omit ; //@sic new CA-Enh test development sic@
    var SubFrameTiming_Type v_TimingInfo;
    var SubFrameTiming_Type v_TimingInfo2;
    //@sic R5w150205 sic@
    var template (omit)Pcell_Mode_Type v_Pcell_Mode := omit;
    var EUTRA_FDD_TDD_CellInfo_Type v_SCellFDD_TDDInfo := f_EUTRA_CellInfo_GetFDD_TDD_Info (p_SCellId);
    var EUTRA_FDD_TDD_CellInfo_Type v_PcellFDD_TDDInfo := f_EUTRA_CellInfo_GetFDD_TDD_Info(p_PCellId);
    if (v_PcellFDD_TDDInfo.FDD_TDD != v_SCellFDD_TDDInfo.FDD_TDD) //Scell mode is different than Pcell
    {
        if (v_PcellFDD_TDDInfo.FDD_TDD == FDD) //Pcell is FDD
        {
           v_Pcell_Mode := cs_Pcell_Mode_FDD
        } else //Pcell is TDD
        {
            v_Pcell_Mode := cs_Pcell_Mode_TDD (v_PcellFDD_TDDInfo.TDD_SubframeAssignment);
        }
    };
    // Step 1: Configure PCell as Pcell or normal cell
    f_EUTRA_SS_CommonCellConfig(p_PCellId, cas_PCellConfig_REQ(p_PCellId, cs_TimingInfo_Now, {p_SCellId}));
    if (isvalue(v_RRCConnectionReconfigurationIEs.radioResourceConfigDedicated.mac_MainConfig.explicitValue.mac_MainConfig_v1020.sCellDeactivationTimer_r10)) {
      v_DeactTimer := v_RRCConnectionReconfigurationIEs.radioResourceConfigDedicated.mac_MainConfig.explicitValue.mac_MainConfig_v1020.sCellDeactivationTimer_r10;
    }
    v_NoUL_SCell := f_EUTRA_CellInfo_GetNoUL_Frequency(p_SCellId);
    v_SCellIndex := v_RRCConnectionReconfigurationIEs.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.sCellToAddModList_r10[0].sCellIndex_r10;
    if (not v_NoUL_SCell) {
      // Step 2: Configure C-RNTI based contention resolution in S-cell
      f_EUTRA_SS_ConfigRachProcedure(p_SCellId, tsc_C_RNTI_Def, cs_RachProcedureConfig_CRNTI(f_EUTRA_CellInfo_GetRAR_TA(p_SCellId),
                                                                                             f_EUTRA_CellInfo_GetDL_ChBandwidth(p_SCellId)));
    }
    // Save new C-RNTI in cell configuration for p_SCellId
    f_EUTRA_CellInfo_SetC_RNTI(p_SCellId, tsc_C_RNTI_Def);
    // Step 3: configure DRX and MeasGap at the S cell (if necessary)
    if (f_EUTRA_CellInfo_DrxIsConfigured(p_PCellId)) {
      v_DrxCtrl := f_EUTRA_CellInfo_GetDrxCtrl(p_PCellId);
      f_EUTRA_SS_DrxCtrlConfig(p_SCellId, v_DrxCtrl);
    }
    if (v_MeasGapConfigIsPresent) {
      f_EUTRA_SS_MeasGapCtrlConfig(p_SCellId, cs_MeasGapCtrl_Config(valueof(p_MeasGapConfig)));   // valueof to avoid compiler warnings: ok as v_MeasConfig.measGapConfig is checked to be present
    }
    if (ispresent (v_RRCConnectionReconfigurationIEs.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.sCellToAddModList_r10[0].radioResourceConfigDedicatedSCell_r10.mac_MainConfigSCell_r11)) {
      v_STAG_Id := v_RRCConnectionReconfigurationIEs.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.sCellToAddModList_r10[0].radioResourceConfigDedicatedSCell_r10.mac_MainConfigSCell_r11.stag_Id_r11;
    }
    v_TimingInfo := f_EUTRA_GetNextSendOccasion(p_PCellId);
    v_TimingInfo2 := f_EUTRA_TimingInfoAdd(v_TimingInfo,5); // 5 ms later
    // Step 6: Transmit RRCConnectionReconfiguration/HO command to UE
    SRB.send(cas_SRB1_RrcPdu_REQ(p_PCellId, cs_TimingInfo(v_TimingInfo.SFN.Number, v_TimingInfo.Subframe.Number), p_RRC_ConnRecongMsg));
    // Configure Pcell for TM change
    if (ischosen(v_RRCConnectionReconfigurationIEs.radioResourceConfigDedicated.physicalConfigDedicated.antennaInfo_r10.explicitValue_r10)) {
      f_EUTRA_AntennaInfoR10_Config(p_PCellId,
                                    cs_TimingInfo(v_TimingInfo2.SFN.Number, v_TimingInfo2.Subframe.Number),
                                    v_RRCConnectionReconfigurationIEs.radioResourceConfigDedicated.physicalConfigDedicated.antennaInfo_r10.explicitValue_r10);
    }
    // Step 4: Configure SCell as Scell
    f_EUTRA_SS_CommonCellConfig(p_SCellId, cas_SCellConfig_REQ(p_SCellId, cs_TimingInfo(v_TimingInfo2.SFN.Number, v_TimingInfo2.Subframe.Number), cs_SCellConfig(p_PCellId, v_SCellIndex, p_Scell_Capability, v_DeactTimer,-,v_STAG_Id,v_Pcell_Mode))); //@sic R5w150205 sic@
    if (not v_NoUL_SCell) {
      // Step 5: configure Stop UL grant configuration on S cell
      f_EUTRA_SS_CommonCellConfig(p_SCellId, cas_ULGrantAllocation_REQ(p_SCellId, cs_TimingInfo(v_TimingInfo2.SFN.Number, v_TimingInfo2.Subframe.Number), p_DciUlInfo_SCell, cs_PUCCH_Synch_None, p_UL_GrantConfig_SCell));
    }
    // Configure Scell for TM change
    //@sic R5s150406 sic@
    if (ispresent(v_RRCConnectionReconfigurationIEs.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.sCellToAddModList_r10[0].radioResourceConfigDedicatedSCell_r10.physicalConfigDedicatedSCell_r10.nonUL_Configuration_r10.antennaInfo_r10)) {
      f_EUTRA_AntennaInfoR10_Config(p_SCellId, cs_TimingInfo(v_TimingInfo2.SFN.Number, v_TimingInfo2.Subframe.Number), v_RRCConnectionReconfigurationIEs.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.sCellToAddModList_r10[0].radioResourceConfigDedicatedSCell_r10.physicalConfigDedicatedSCell_r10.nonUL_Configuration_r10.antennaInfo_r10);
    }
    // Step 7: Receive RRCConnectionReconfiguration Complete
    SRB.receive(car_SRB1_RrcPdu_IND(p_PCellId, cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));
    // No RACH configuration as Already configured for C-RNTI based contention
  }
...
  function f_EUTRA_ConfigPcell_AddScells(EUTRA_CellId_Type p_PCellId,
                                         EUTRA_CellIdList_Type p_SCellIdAddList := tsc_EUTRA_NullCellIdList,
                                         template (value) RachProcedureConfig_Type p_RachProcedureConfig,
                                         C_RNTI p_C_RNTI := tsc_C_RNTI_Def,
                                         template (omit) MeasGapConfig p_MeasGapConfig := omit, //@sic R5s130503 sic@
                                         boolean p_CnfFlag := tsc_NoCnfReq)
    runs on EUTRA_PTC
  {
    var boolean v_MeasGapConfigIsPresent := isvalue(p_MeasGapConfig);
    var boolean v_NoUL_SCell; //@sic R5s140083 sic@
    var DrxCtrl_Type v_DrxCtrl;
    var integer i;
    // Step 1: Configure PCell as Pcell or normal cell
    if (lengthof(p_SCellIdAddList) > 0) { // one or more SCell Addition
      f_EUTRA_SS_CommonCellConfig(p_PCellId, cas_PCellConfig_REQ(p_PCellId, cs_TimingInfo_Now, p_SCellIdAddList, p_CnfFlag));
    }
...
  function f_EUTRA_508RRC_AddModRel_Scell_Common(EUTRA_CellId_Type                         p_PCellId,
                                                 EUTRA_CellIdList_Type                     p_SCellIdAddList := tsc_EUTRA_NullCellIdList,
                                                 EUTRA_CellIdList_Type                     p_SCellIdDelList := tsc_EUTRA_NullCellIdList,
                                                 template (value) DL_DCCH_Message          p_RRC_ConnRecongMsg,
                                                 template (value) Scell_Capability_Type    p_Scell_Capability := UL_DL,
                                                 template (value) RachProcedureConfig_Type p_RachProcedureConfig,
                                                 C_RNTI                                    p_C_RNTI := tsc_C_RNTI_Def,
                                                 template (omit) MeasGapConfig             p_MeasGapConfig := omit, //@sic R5s130503 sic@
                                                 template (omit)  DciUlInfo_Type           p_DciUlInfo_SCell := cs_DciInfo_CcchDcchDtchUL_Def,
                                                 template (value) UL_GrantConfig_Type      p_UL_GrantConfig_SCell := cs_UL_GrantConfig_Stop,
                                                 boolean                                   p_PcelltoNormalCell := false)   //Configure Pcell to Normal cell if p_SCellIdAddList is empty
    runs on EUTRA_PTC
  {
    var integer i;
    var SubFrameTiming_Type v_TimingInfo;
    var SubFrameTiming_Type v_TimingInfo2;
    var EUTRA_FDD_TDD_CellInfo_Type v_FDD_TDDInfo_PCell := f_EUTRA_CellInfo_GetFDD_TDD_Info(p_PCellId);
    var boolean v_CnfFlag := tsc_NoCnfReq;
    f_EUTRA_ConfigPcell_AddScells(p_PCellId,
                                  p_SCellIdAddList,
                                  //p_RRC_ConnRecongMsg,
                                  p_RachProcedureConfig,
                                  p_C_RNTI,
                                  p_MeasGapConfig);
    //@sic R5s140376 sic@
    v_TimingInfo := f_EUTRA_GetNextSendOccasion(p_PCellId);
    v_TimingInfo2 := f_EUTRA_TimingInfoAdd(v_TimingInfo, 5); // 5 ms later
    // Step 6: Transmit RRCConnectionReconfiguration/HO command to UE
    SRB.send(cas_SRB1_RrcPdu_REQ(p_PCellId, cs_TimingInfo(v_TimingInfo.SFN.Number, v_TimingInfo.Subframe.Number), p_RRC_ConnRecongMsg));
    //@sic R5s150070 sic@
...
After change

...
  function f_EUTRA_508RRC_AddModRel_Scell_MIMO(EUTRA_CellId_Type p_PCellId,
                                               EUTRA_CellId_Type p_SCellId,
                                               template (value) DL_DCCH_Message p_RRC_ConnRecongMsg,
                                               template (value) Scell_Capability_Type p_Scell_Capability := UL_DL,
                                               template (omit) MeasGapConfig p_MeasGapConfig := omit,
                                               template (omit)  DciUlInfo_Type p_DciUlInfo_SCell := cs_DciInfo_CcchDcchDtchUL_Def,
                                               template (value) UL_GrantConfig_Type  p_UL_GrantConfig_SCell := cs_UL_GrantConfig_Def)
    runs on EUTRA_PTC
  {
    var boolean v_MeasGapConfigIsPresent := isvalue(p_MeasGapConfig);
    var DrxCtrl_Type v_DrxCtrl;
    var RRCConnectionReconfiguration_r8_IEs v_RRCConnectionReconfigurationIEs := valueof(p_RRC_ConnRecongMsg.message_.c1.rrcConnectionReconfiguration.criticalExtensions.c1.rrcConnectionReconfiguration_r8);
    var SCellIndex_r10 v_SCellIndex := 1;
    var MAC_MainConfig_ScellDeactivationTimer_Type v_DeactTimer := rf32;
    var boolean v_NoUL_SCell ; //@sic R5s140083 sic@
    var template (omit) STAG_Id_r11 v_STAG_Id := omit ; //@sic new CA-Enh test development sic@
    var SubFrameTiming_Type v_TimingInfo;
    var SubFrameTiming_Type v_TimingInfo2;
    //@sic R5w150205 sic@
    var template (omit)Pcell_Mode_Type v_Pcell_Mode := omit;
    var EUTRA_FDD_TDD_CellInfo_Type v_SCellFDD_TDDInfo := f_EUTRA_CellInfo_GetFDD_TDD_Info (p_SCellId);
    var EUTRA_FDD_TDD_CellInfo_Type v_PcellFDD_TDDInfo := f_EUTRA_CellInfo_GetFDD_TDD_Info(p_PCellId);
    if (v_PcellFDD_TDDInfo.FDD_TDD != v_SCellFDD_TDDInfo.FDD_TDD) //Scell mode is different than Pcell
    {
        if (v_PcellFDD_TDDInfo.FDD_TDD == FDD) //Pcell is FDD
        {
           v_Pcell_Mode := cs_Pcell_Mode_FDD
        } else //Pcell is TDD
        {
            v_Pcell_Mode := cs_Pcell_Mode_TDD (v_PcellFDD_TDDInfo.TDD_SubframeAssignment);
        }
    };
    v_TimingInfo := f_EUTRA_GetNextSendOccasion(p_PCellId);
    v_TimingInfo2 := f_EUTRA_TimingInfoAdd(v_TimingInfo,5); // 5 ms later
    // Step 1: Configure PCell as Pcell or normal cell
    f_EUTRA_SS_CommonCellConfig(p_PCellId, cas_PCellConfig_REQ(p_PCellId, cs_TimingInfo(v_TimingInfo2.SFN.Number, v_TimingInfo2.Subframe.Number), {p_SCellId}));
    if (isvalue(v_RRCConnectionReconfigurationIEs.radioResourceConfigDedicated.mac_MainConfig.explicitValue.mac_MainConfig_v1020.sCellDeactivationTimer_r10)) {
      v_DeactTimer := v_RRCConnectionReconfigurationIEs.radioResourceConfigDedicated.mac_MainConfig.explicitValue.mac_MainConfig_v1020.sCellDeactivationTimer_r10;
    }
    v_NoUL_SCell := f_EUTRA_CellInfo_GetNoUL_Frequency(p_SCellId);
    v_SCellIndex := v_RRCConnectionReconfigurationIEs.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.sCellToAddModList_r10[0].sCellIndex_r10;
    if (not v_NoUL_SCell) {
      // Step 2: Configure C-RNTI based contention resolution in S-cell
      f_EUTRA_SS_ConfigRachProcedure(p_SCellId, tsc_C_RNTI_Def, cs_RachProcedureConfig_CRNTI(f_EUTRA_CellInfo_GetRAR_TA(p_SCellId),
                                                                                             f_EUTRA_CellInfo_GetDL_ChBandwidth(p_SCellId)));
    }
    // Save new C-RNTI in cell configuration for p_SCellId
    f_EUTRA_CellInfo_SetC_RNTI(p_SCellId, tsc_C_RNTI_Def);
    // Step 3: configure DRX and MeasGap at the S cell (if necessary)
    if (f_EUTRA_CellInfo_DrxIsConfigured(p_PCellId)) {
      v_DrxCtrl := f_EUTRA_CellInfo_GetDrxCtrl(p_PCellId);
      f_EUTRA_SS_DrxCtrlConfig(p_SCellId, v_DrxCtrl);
    }
    if (v_MeasGapConfigIsPresent) {
      f_EUTRA_SS_MeasGapCtrlConfig(p_SCellId, cs_MeasGapCtrl_Config(valueof(p_MeasGapConfig)));   // valueof to avoid compiler warnings: ok as v_MeasConfig.measGapConfig is checked to be present
    }
    if (ispresent (v_RRCConnectionReconfigurationIEs.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.sCellToAddModList_r10[0].radioResourceConfigDedicatedSCell_r10.mac_MainConfigSCell_r11)) {
      v_STAG_Id := v_RRCConnectionReconfigurationIEs.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.sCellToAddModList_r10[0].radioResourceConfigDedicatedSCell_r10.mac_MainConfigSCell_r11.stag_Id_r11;
    }
    // Step 6: Transmit RRCConnectionReconfiguration/HO command to UE
    SRB.send(cas_SRB1_RrcPdu_REQ(p_PCellId, cs_TimingInfo(v_TimingInfo.SFN.Number, v_TimingInfo.Subframe.Number), p_RRC_ConnRecongMsg));
    // Configure Pcell for TM change
    if (ischosen(v_RRCConnectionReconfigurationIEs.radioResourceConfigDedicated.physicalConfigDedicated.antennaInfo_r10.explicitValue_r10)) {
      f_EUTRA_AntennaInfoR10_Config(p_PCellId,
                                    cs_TimingInfo(v_TimingInfo2.SFN.Number, v_TimingInfo2.Subframe.Number),
                                    v_RRCConnectionReconfigurationIEs.radioResourceConfigDedicated.physicalConfigDedicated.antennaInfo_r10.explicitValue_r10);
    }
    // Step 4: Configure SCell as Scell
    f_EUTRA_SS_CommonCellConfig(p_SCellId, cas_SCellConfig_REQ(p_SCellId, cs_TimingInfo(v_TimingInfo2.SFN.Number, v_TimingInfo2.Subframe.Number), cs_SCellConfig(p_PCellId, v_SCellIndex, p_Scell_Capability, v_DeactTimer,-,v_STAG_Id,v_Pcell_Mode))); //@sic R5w150205 sic@
    if (not v_NoUL_SCell) {
      // Step 5: configure Stop UL grant configuration on S cell
      f_EUTRA_SS_CommonCellConfig(p_SCellId, cas_ULGrantAllocation_REQ(p_SCellId, cs_TimingInfo(v_TimingInfo2.SFN.Number, v_TimingInfo2.Subframe.Number), p_DciUlInfo_SCell, cs_PUCCH_Synch_None, p_UL_GrantConfig_SCell));
    }
    // Configure Scell for TM change
    //@sic R5s150406 sic@
    if (ispresent(v_RRCConnectionReconfigurationIEs.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.sCellToAddModList_r10[0].radioResourceConfigDedicatedSCell_r10.physicalConfigDedicatedSCell_r10.nonUL_Configuration_r10.antennaInfo_r10)) {
      f_EUTRA_AntennaInfoR10_Config(p_SCellId, cs_TimingInfo(v_TimingInfo2.SFN.Number, v_TimingInfo2.Subframe.Number), v_RRCConnectionReconfigurationIEs.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.sCellToAddModList_r10[0].radioResourceConfigDedicatedSCell_r10.physicalConfigDedicatedSCell_r10.nonUL_Configuration_r10.antennaInfo_r10);
    }
    // Step 7: Receive RRCConnectionReconfiguration Complete
    SRB.receive(car_SRB1_RrcPdu_IND(p_PCellId, cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));
    // No RACH configuration as Already configured for C-RNTI based contention
  }
... 
  function f_EUTRA_ConfigPcell_AddScells(EUTRA_CellId_Type p_PCellId,
                                         EUTRA_CellIdList_Type p_SCellIdAddList := tsc_EUTRA_NullCellIdList,
                                         template (value) RachProcedureConfig_Type p_RachProcedureConfig,
                                         C_RNTI p_C_RNTI := tsc_C_RNTI_Def,
                                         template (omit) MeasGapConfig p_MeasGapConfig := omit, //@sic R5s130503 sic@
                                         template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,
                                         boolean p_CnfFlag := tsc_NoCnfReq)
    runs on EUTRA_PTC
  {
    var boolean v_MeasGapConfigIsPresent := isvalue(p_MeasGapConfig);
    var boolean v_NoUL_SCell; //@sic R5s140083 sic@
    var DrxCtrl_Type v_DrxCtrl;
    var integer i;
    // Step 1: Configure PCell as Pcell or normal cell
    if (lengthof(p_SCellIdAddList) > 0) { // one or more SCell Addition
      f_EUTRA_SS_CommonCellConfig(p_PCellId, cas_PCellConfig_REQ(p_PCellId, p_TimingInfo, p_SCellIdAddList, p_CnfFlag));
    }
...
  function f_EUTRA_508RRC_AddModRel_Scell_Common(EUTRA_CellId_Type                         p_PCellId,
                                                 EUTRA_CellIdList_Type                     p_SCellIdAddList := tsc_EUTRA_NullCellIdList,
                                                 EUTRA_CellIdList_Type                     p_SCellIdDelList := tsc_EUTRA_NullCellIdList,
                                                 template (value) DL_DCCH_Message          p_RRC_ConnRecongMsg,
                                                 template (value) Scell_Capability_Type    p_Scell_Capability := UL_DL,
                                                 template (value) RachProcedureConfig_Type p_RachProcedureConfig,
                                                 C_RNTI                                    p_C_RNTI := tsc_C_RNTI_Def,
                                                 template (omit) MeasGapConfig             p_MeasGapConfig := omit, //@sic R5s130503 sic@
                                                 template (omit)  DciUlInfo_Type           p_DciUlInfo_SCell := cs_DciInfo_CcchDcchDtchUL_Def,
                                                 template (value) UL_GrantConfig_Type      p_UL_GrantConfig_SCell := cs_UL_GrantConfig_Stop,
                                                 boolean                                   p_PcelltoNormalCell := false)   //Configure Pcell to Normal cell if p_SCellIdAddList is empty
    runs on EUTRA_PTC
  {
    var integer i;
    var SubFrameTiming_Type v_TimingInfo;
    var SubFrameTiming_Type v_TimingInfo2;
    var EUTRA_FDD_TDD_CellInfo_Type v_FDD_TDDInfo_PCell := f_EUTRA_CellInfo_GetFDD_TDD_Info(p_PCellId);
    var boolean v_CnfFlag := tsc_NoCnfReq;
    //@sic R5s140376 sic@
    v_TimingInfo := f_EUTRA_GetNextSendOccasion(p_PCellId);
    v_TimingInfo2 := f_EUTRA_TimingInfoAdd(v_TimingInfo, 5); // 5 ms later
    f_EUTRA_ConfigPcell_AddScells(p_PCellId,
                                  p_SCellIdAddList,
                                  //p_RRC_ConnRecongMsg,
                                  p_RachProcedureConfig,
                                  p_C_RNTI,
                                  p_MeasGapConfig,
                                  cs_TimingInfo(v_TimingInfo2.SFN.Number, v_TimingInfo2.Subframe.Number));
    // Step 6: Transmit RRCConnectionReconfiguration/HO command to UE
    SRB.send(cas_SRB1_RrcPdu_REQ(p_PCellId, cs_TimingInfo(v_TimingInfo.SFN.Number, v_TimingInfo.Subframe.Number), p_RRC_ConnRecongMsg));
    //@sic R5s150070 sic@

...
